

UMBC

AN HONORS UNIVERSITY IN MARYLAND

Department of Computer Science and Electrical Engineering

CMPE 415

Working with Operands and Variables

Prof. Ryan Robucci

<http://www.asic-world.com/verilog/operators1.html>

<http://www.asic-world.com/verilog/operators2.html>

Data Types: Nets and Variables

Nets: Provide structural connectivity, its value is determined by connection or continuous assignment

Variables: Provide connectivity as well as ability to be assigned in a procedural block

- reg: stores logic value
 - integer: Supports Computation
 - real: stores values as real numbers
 - time: stores time as u64
 - realtime: stores time as real
- Variables were originally called registers, but since they are not actually always implemented as registers in hardware the name was changed for Verilog 2001. Take note when reading older texts.

reg

- A variable type which can be used to abstract the hardware storage element since it remember values set in procedural code or sequential primitives –
 - but it may also represent a node in a hardware circuit who's value is continuously set by combinatorial logic
- It is assigned a value in Verilog by one of
 - procedural statement
 - user-defined seq. primitive
 - task or function
- May not connect to output (or inout) ports of instantiated modules or Verilog primitives
- May not be assigned value using a continuous assignment

reg vectors

- reg variables are commonly used as vectors.

```
reg [msbindex:lsbindex] <varname>;
```

indexes may be negative

Two vector declarations with different bit-significance ordering.
msb-first and lsb-first

```
reg [31:0] regA;  The most-significant bits are implied  
reg [0:31] regB; to be regA[31] and regB[0]
```

```
regA = regB;      sets  regA[31] to regB[ 0]  
                  regA[30] to regB[ 1]  ....  
                  regA[ 0] to regB[31]
```

Bits may be read and assigned one at a time using the indexing operator []:

```
regA[31] = regA[0];
```

reg vectors

Multiple bits may be read and assigned using the indexing operator []:

```
regA[31:24] = regA[7:0];
```

Multiple bits may be read and assigned using the indexed part-select syntax:

```
regA[ 0 +: 8] // regA[ 7 : 0]
regA[15 -: 8] // regA[15 : 8]
regB[ 0 +: 8] // regB[0 : 7]
regB[15 -: 8] // regB[8 :15]
```

In general, I highly recommend always using msb-first convention as the following statement can be confusing:

```
regA[ 0 +: 8] = regB[ 0 +: 8];
```

Is the same as `regA[7 : 0] = regB[0 : 7];`

Signed reg vectors and wire vecors

- reg variables and wire nets may be signed. These support signed operations like integers (integers are signed by default).
- signed reg [msb:lsb] <varname>;
signed wire [msb:lsb] <varname>;

Ex:

```
signed reg [7:0] regA;  
signed wire [7:0] wireA;
```

These support values -128 to 127 and the signed modifier determines different behaviors for some operations that support signed numbers.

Convention for the slides herein

Slides in this class may include several examples where we won't want to see all declarations crowding the slides.

Assume prefixes `ur1,sr8`, `ur8,ur16,ur32`, etc. denote unsigned variables of obvious lengths 8,16,32, etc.

Assume prefixes `sr1,sr4,sr8,sr16,sr32`, etc. denote signed variables of obvious lengths 1,4,8,16,32, etc.

Examples:

```
reg [7:0] ur8a;  
signed reg [15:0] sr15a;
```

Assume prefixes `uw1,sw8`, `uw8,uw16,uw32`, etc. denote unsigned nets of obvious lengths 8,16,32, etc.

Assume prefixes `sw1,sw4,sw8,sw16,sw32`, etc. denote signed nets of obvious lengths 1,4,8,16,32, etc.

Examples:

```
wire [7:0] uw8a;  
signed reg [15:0] sw15a;
```

Integers

- Generally act as a 32-bit signed value, but can be larger
- If an integer is used where a `reg[31:0]` is required, providing the integer can cause an implicit casting to unsigned – this interpretation of a signed storage as an unsigned can be a disaster.
- Example declaration:

```
integer int1, int2;  
integer intX=100;  
integer intY=10;
```


Integers as throw away variable

- In general you may use integers for named constants and as temporary throw-away variables in your code where computations assigned to integers are computed strictly at synthesis
- They are dangerous to use as registers (where we use `<=`) and caution for other use is required.

```
integer intX=100;  
integer intY=10;  
integer intO=5;  
integer intZ;  
reg [31:0] u32q;
```

```
always @(posedge clk)
```

```
    intZ = intX/intY; ← Can be precomputed
```

```
    u32q <= u32q + intZ; ← final result is reg and designer
```

```
end
```

determined that its size will hold the value in intZ this time

Unsigned Sized Literals

- When specifying numerical literals, a negative sign can denote a two's complement operation (performs one's complement then add 1) on an unsigned sized value and still result in an unsigned value (a value that operands treat as unsigned)
- Ex: `-'d8` is a 32-bit unsigned value storing a signed representation of (-8)
This means that operations don't know to use signed arithmetic on it even though we might like to think of the bits as representing an signed representation
 - `(-'d8)/2` does not produce “-4”
since `-'d8` is unsigned and thus unsigned division is performed which does not preserve sign
 - `-'d8)/2` does produce “-4”

Integer and Unsigned Sized Literals

```
integer int1,int2,int3,int4,int5,int6,int7;
initial begin
  int1=  -8;          // 1111 1111 1111 1111 1111 1111 1111 1111 1000
  int2=  -'d8;       // 1111 1111 1111 1111 1111 1111 1111 1111 1000
  int3=  -'d8/2;     // 0111 1111 1111 1111 1111 1111 1111 1111 1100
  int4=  -8/2;       // 1111 1111 1111 1111 1111 1111 1111 1111 1100
  int5=  int1/2;     // 1111 1111 1111 1111 1111 1111 1111 1111 1100
  int6=  int2/2;     // 1111 1111 1111 1111 1111 1111 1111 1111 1100
  int7=  -('d8/2);   // 1111 1111 1111 1111 1111 1111 1111 1111 1100
  int8=  (-'d8)/2;   // 0111 1111 1111 1111 1111 1111 1111 1111 1100
end
```

Signed Integer and Sized Literals

- One solution is the new signed sized literals using `s` in front of the format specifier to provide a signed type

- Ex: `-'sd8` is a 32-bit **signed** value storing a signed representation of `(-8)`

This means that operations **do** know to use signed arithmetic

- `(-'sd8)/2` **does** produce `"-4"`
since `(-'sd8)` is **signed** and thus **signed** division is performed which **does** preserve sign
- `-('sd8)/2` does produce `"-4"`
- Ex:
 - `16'sh00A0; //16-bit signed +10`
 - `-16'sh00A0; //16-bit signed -10`
 - `16'shFFFF; //16-bit signed -1`

Signed Integer and Sized Literals

```
integer int1,int2,int3,int4,int5,int6,int7;
initial begin
  int1=  -8;           // 1111 1111 1111 1111 1111 1111 1111 1111 1000
  int2=  -'sd8;       // 1111 1111 1111 1111 1111 1111 1111 1111 1000
  int3=  -'sd8/2;     // 1111 1111 1111 1111 1111 1111 1111 1111 1100
  int4=  -8/2;        // 1111 1111 1111 1111 1111 1111 1111 1111 1100
  int5=  int1/2;      // 1111 1111 1111 1111 1111 1111 1111 1111 1100
  int6=  int2/2;      // 1111 1111 1111 1111 1111 1111 1111 1111 1100
  int7=  -('sd8/2);   // 1111 1111 1111 1111 1111 1111 1111 1111 1100
  int8=  (-'sd8)/2;   // 1111 1111 1111 1111 1111 1111 1111 1111 1100
end
```

Reals

- Reals generally treated as double, 64 bit elements
- Literals may be provided in exponential (e.g. 1.2e-3) or basic decimal format (.34)
- No bit access
- No direct passing through ports of primitives or modules
- Use \$realtobits, \$bitstoreal, \$rtoi (truncate), and \$itor to convert to 64-bit binary and integers
 - May be used to pass reals through 64-bit ports that are declared as follows:

```
module (output wire [63:0] realIn,  
        input wire [63:0] realOut);  
real tmp;  
tmp = $bitstoreal(realIn);  
realOut = $realtobits(tmp);
```

time, realtime, and `timescale

- time is a u64
- realtime is a real
- Like integer and real, these may not be used in a module port or be an output or input of a primitive
- Time is defined according to the time_unit and base provided by

`timescale time_unit base / precision base

- time_unit is the time step represented by **#1.0**
- base is s ms us ns ps or fs
- precision is the rounding precision of the simulation
- Example:

```
`timescale 1ns/1ps  
sets #1.0 to mean 1 ns delay
```

Checking `timescale

- If you want to print the timescale used in your simulation, you can print it using the following in your testbench or in other files.

```
initial $prinntimescale;
```

Prints the following when place in module DUT_tb:

```
Timescale of (DUT_tb) is 1ns/1ps.
```

- Checking other modules in the hierarchy can be done by providing the hierarchical name of the module:

```
initial $prinntimescale(DUT_tb.DUT.I4);
```


Time-related system tasks and functions

- Related system functions:

- **\$time** is a system function which provides an integer

- **\$stime** provides a truncated 32-bit time

- **\$realtime** provides a real (64-bit) time according

- In a format string for a system task like **\$display**, use “%0t” to print time:

- **\$display**("time is %0t",**\$time**);

- %t follows a default formatting that which can be redefined using the system task

- **\$timeformat**(unit, precision, suffix, min_field_width);

- **unit** is a value from 0 to -15 representing seconds to femtoseconds

- **precision** represents the number of decimal points to display

- **suffix** is a string to append to what is printed (e.g. “ns”)

- **min_field_width** represents the minimum number of characters to use to print time

Arrays

- A 2D bit memory can be defined with reg array

Example:

```
reg [31:0] cache_memory [1023:0]
```

Defines 1024 32-bit entries

- Integer(32-bit) array

```
integer cache_memory [1023:0];
```

- Real array

```
real sin_table[1023:0];
```

- An array of time can be created:

```
time T[1:00];
```

- Multi-dimensional arrays are allowed

- Example:

- ```
reg [31:0] cache_memory [3:0][1023:0]
```

- 16 banks of 1024 words of length 32 bits each,

- ```
cache_memory [15][1023][0]
```

 is the lsb of the last work

Strings

- no built in char type,
 - just reserve 8 bits per char in reg declaration

```
parameter numchars = 7;  
reg [((8*numchars)-1):0] my_string;  
my_string = "Hi";
```

Result is padded with 0s on left and no trailing 0:

```
my_string = 0 0 0 0 0 'H' 'I'
```

Constants may be stored and recalled with parameters

Parameters are a good way to store constants, but be mindful of the type being assigned.

```
parameter a=31; //int
```

```
parameter a=32,b=31; //ints
```

```
parameter byte_size=8, byte_max=byte_size-1; //int
```

```
parameter a =6.22; //real
```

```
parameter delay = (min_delay + max_delay) /2 //real
```

```
parameter initial_state = 8'b1001_0110; //reg
```

Arithmetic Operators

- Binary: +, -, *, /, % (the modulus operator)
- Unary: +, - (This is used to specify the sign)
- Integer division truncates any fractional part
- The result of a modulus operation takes the sign of the first operand
- Unlike bitwise operators, if any operand bit value is the unknown value x, then the entire result value is x (e.g. $x * 0 \rightarrow x$)
- Register data types are used as unsigned values (Negative numbers are stored in two's complement form)

$$5 + 10 = 15$$

$$5 - 10 = -5$$

$$10 - 5 = 5$$

$$10 * 5 = 50$$

$$10 / 5 = 2$$

$$10 / -5 = -2$$

$$10 \% 3 = 1$$

$$+5 = 5$$

$$-5 = -5$$

Reduction Operators

- Unary operators (operate on one operand)
 - Produce single bit result
 - Operations with x and z bits may be resolvable (e.g x and 0 = 0)

Symbol	Operator
$\&$	and
$ $	or
$\sim $	nor
\wedge	xor
$\sim\wedge, \wedge\sim$	xnor

```

& 2'b1z   → x
& 2'b1x   → x
& 2'b0z   → 0
& 2'b0x   → 0
& 0       → 0
& 1       → 0
& -1      → 1
    
```

$\& (010101) \rightarrow 0$ Reduction And

$| (010101) \rightarrow 1$ Reduction Or

$\& (010x10) \rightarrow 0$ Reduction And

$| (010x10) \rightarrow 1$ Reduction Or

Logical Operators

Symbol	Operator
!	negation
&&,	and,or
==,! =	Logical equality
===,! ===	Case equality

- ===,! === compare bit matching including x and z
- ==,! = produce x if any x or z exists
- left zero padding performed as needed
- Signed and unsigned integers and reals are converted to binary and treated as unsigned
- && look for two non-zero valid words
- || looks for one non-zero, both valid words
- What about a non-valid bits
 - (3'b110 && 3'b11x) → (1 && 0) → FALSE=0
 - (3'b110 & 3'b11x) → (3'b110) → TRUE=1

Operator	Description
<code>a === b</code>	a equal to b, including x and z (Case equality)
<code>a !== b</code>	a not equal to b, including x and z (Case inequality)
<code>a == b</code>	a equal to b, result may be unknown (logical equality)
<code>a != b</code>	a not equal to b, result may be unknown (logical equality)

- Operands are compared bit by bit, with zero filling if the two operands do not have the same length
- Result is 0 (false) or 1 (true)
- For the `==` and `!=` operators, the result is x, if either operand contains an x or a z
- For the `===` and `!==` operators, bits with x and z are included in the comparison and must match for the result to be true

Note : The result is always 0 or 1 unless result is unknown and x results

Equality Examples

4'bx001 === 4'bx001 → 1

4'bx0x1 === 4'bx001 → 0

4'bz0x1 === 4'bz0x1 → 1

4'bz0x1 === 4'bz001 → 0

4'bx0x1 !== 4'bx001 → 1

4'bz0x1 !== 4'bz001 → 1

5 == 10 → 0

5 == 5 → 1

5 != 5 → 0

5 != 6 → 1

Bit-Wise Operators

- Bit-wise operators on pairs of operands
 - output same size as inputs
 - shorter binary operands are zero-padded on the left to the longer
 - these operators imply gates

Symbol	Operator
~	not/compliment
&	and
	or
^	xor
~^	xnor

```

~4'b0001           = 1110
~4'bx001          = x110
~4'bz001          = x110
4'b0001 & 4'b1001 = 0001
4'b1001 & 4'bx001 = x001
4'b1001 & 4'bz001 = x001
4'b0001 | 4'b1001 = 1001
4'b0001 | 4'bx001 = x001
4'b0001 | 4'bz001 = x001
4'b0001 ^ 4'b1001 = 1000
4'b0001 ^ 4'bx001 = x000
4'b0001 ^ 4'bz001 = x000
4'b0001 ~^ 4'b1001 = 0111
4'b0001 ~^ 4'bx001 = x111
4'b0001 ~^ 4'bz001 = x111
    
```

- Note x (z is treated same as x) is not always “contagious” if the operation makes the result unambiguous
 - $x \& 1'b0 \rightarrow 0$ and $x | 1'b1 \rightarrow 1$

Logical Operators

Operator	Description
!	Logical negation
&&	Logical and
	Logical or

1'b1 && 1'b1 → 1
1'b1 && 1'b0 → 0
1'b1 && 1'bx → x
1'b0 && 1'bx → 0
1'b1 || 1'b0 → 1
1'b0 || 1'b0 → 0
1'b0 || 1'bx → x
1'b1 || 1'bx → 1

- Expressions connected by && and || are evaluated from left to right
(con1 && cond2 || cond3)
- Evaluation stops as soon as the result is known
- The result is a scalar value:
 - 0 if the relation is false
 - 1 if the relation is true
 - **x if any of the operands has x (unknown) bits**

Logic Negation, !

- Creates inverse of logic result
- For binary vectors,
 - Searches vector left to right
 - if zero found, move right
 - if 1 found, stop evaluation result is 0
 - if x or z found, stop evaluation result is x
 - if end found, result is 1
- For integers,
 - returns 1 for 0 and 0 otherwise

!	1'b1	=	0
!	1'b0	=	1
!	1'bz	=	x
!	1'bx	=	x
!	2'b00	=	1
!	2'b01	=	0
!	2'b10	=	0
!	2'b11	=	0
!	2'b1z	=	0
!	2'b1x	=	0
!	2'b0z	=	x
!	2'b0x	=	x
!	0	=	1
!	1	=	0
!	(-1)	=	0

Relational Operators

- Nets and registers treated as unsigned words
- If any bit is unknown, the relation is unknown and result is ambiguous x

Symbol	Operator
<	Less than
<=	Less than equal to
>	Greater than
>=	Greater than or equal to

When one or both operands of a relational expression are unsigned, the expression shall be interpreted as a comparison between unsigned values. If the operands are of unequal bit lengths, the smaller operand shall be zero-extended to the size of the larger operand. --IEEE Spec

Implicit Casting Surprises

- Mix of unsigned and signed results in unsigned cast examples for W=32

Constant	Constant	Relation	Evaluation
u8a=0	<code>\$unsigned(0)</code>	<code>==</code>	unsigned
s8a=-1	0	<code><</code>	signed
s8a=-1	<code>\$unsigned(0)</code>	<code>></code>	unsigned
s8a=127	<code>s8b = -127-1</code>	<code>></code>	signed
<code>u8a=\$unsigned(127)</code>	<code>s8b = -127-1</code>	<code><</code>	unsigned
s8a=-1	-2	<code>></code>	signed
<code>u8a=\$unsigned(-1)</code>	-2	<code>></code>	unsigned
s8a=127	<code>\$unsigned(255)</code>	<code><</code>	unsigned
s8a=127	<code>s8a=\$unsigned(255)</code>	<code>></code>	signed

Shift Operators

- Accepts two unsigned binary word operands
- integers are converted to two-complement binary equivalents and treated as unsigned (bit shift with 0-filling)

Symbol	Operator
<<	Left Shift
>>	Right Shift

Conditional Operator

- Z not allowed in conditional_expression
- Zeros are automatically filled for different length operands
- If conditional_expression is ambiguous, both true_ex and false_ex are evaluated and result is calculated on a bitwise basis according to truth table below (same as collision of wires, exists since section doesn't matter if both bits are same)

1'bx?a:b	a=0	a=1	a=x
b=0	0	x	x
b=1	x	1	x
b=x	x	x	x

Concatenation and Repetition Operator

- Concatenation operator can create a single word from two or more operands
- Useful for forming logic buses
- Concatenation follows order given
- Nesting allowed

`{2b'01, 3'b010} → 5'b01010`

`{4{2'b01}} → {2'b01, 2'b01, 2'b01, 2'b01} → 8'b01010101`

`{4'b0001, {{2'b01}, {2'b10}} → 8'b00010110`

No operand may be an unsigned constant since compiler would not be able to size the result

Repetition operator should be used with a constant repetitions:
`<repetitions>{<data>}`

`4{2b'01} → 8'b01010101`

Operator Precedence

Precedence	Operator	Symbols
First	Unary	+ - ! ~
	Multiplication, Division, Modulus	* / %
	Add, Subtract	+ -
	Shift	<< >> (<<< >>>)
	Relational	< <= > >=
		== != === !==
Last	Conditional	? :

Literals (Unsigned)

Number	#Bits	Base	Dec. Equiv.	Stored
2'b10	2	Binary	2	10
3'd5	3	Decimal	5	101
3'o5	3	Octal	5	101
8'o5	s	Octal	5	00000101
8'ha	8	Hex	10	00001010
3'b5	Not Valid!			
3'b01x	3	Binary	-	01 x
12'hx	12	Hex	-	xxxxxxxxxxxx
8'hz	S	Hex	-	zzzzzzzz
8'b0000_0001	8	Binary	1	00000001
8'b0001	8	Binary	1	00000001
8'bx01	8	Binary	-	xxxxxx01
'bz	unsized	Binary	-	z ... z (32 bits)
8'HAD	8	Hex	173	10101101

• Left-most bit is extended as required to fill bits

Working with signed and unsigned reg and wire

Verilog 2001 provides signed reg and wire vectors

Casting two and from signed may be implicit or may be explicit by using

```
$unsigned()    reg_s=$unsigned(reg_u);  
$signed()     reg_u=$unsigned(reg_s);
```

Implicit and Explicit Casting is always a **dumb conversion**

(same as C), they never change bits, just the interpretation of the bits (e.g. -1 is not round to 0 upon conversion to unsigned, it is just reinterpreted as the largest unsigned value) for subsequent operations

Working with signed and unsigned reg and wire

If a mix of signed and unsigned operands are provided to an operator, both operands are first cast to be unsigned (like C)

Signed/Unsigned Casting may be followed with length adjustment
Assignment to a shorter type is always just bit truncation (no smart rounding such as `unsigned(-1) → 0`)

- Error Checking
 - For unsigned, truncation is no problem as long as all the truncated bits are 0.
 - For signed, truncation is no problem as long as all the bits truncated are the same AND they match the surviving msb

Assignment to a longer type is done with either zero or sign extension depending on the type:

- Unsigned types use zero extension
- Signed types use sign extension

Rules for expression bit lengths

A self-determined expression is one in which the length of the result is determined by the length of the operands or in some cases the expression has a predetermined length for the result.

Example: as self-determined expression the result of addition has a length that is the maximum length of the two operands.

Example: a comparison is always a self-determined expression with a 1-bit result

However, addition and other operation expressions may act as a context-determined expression in which the bit length is determined by the context of the expression that contains it.

Addition

Two's complement addition of two numbers where the longest is N-bit can require up to N+1 bits to store a full result

Two's complement addition can only overflow if the signs of the operands are the same

- The error check is to see if the new sign bit is the same as the old

Two's complement subtraction can only under/overflow if the signs of the operands are different

- The error check is to see if the sign bit of the second and third operand are the same and not equal to that of the first

Addition

In this example we see that addition obeys modular arithmetic with a result $ur8y=0$

```
ur8a = 128;
```

```
ur8b = 128;
```

```
ur8y= ur8a+ur8b;
```

In this example we see that the addition is a subexpression paired with an assignment and so the length of the assigned variable sets the context-determined expression operand length of the addition to take on the length of the largest operand, 9-bits. Using zero extension in this case, the addition operands are each extended to 9-bits before addition.

The result is $ur9y=256$

```
ur8a = 128;
```

```
ur8b = 128;
```

```
ur9y= ur8a+ur8b;
```

Self-Determined Expression and Self-Determined Operands

- Some operators always represent a self-determined expression, they have a well-defined bit-length that is independent of the context in which they are used and must be derived directly from the input operand(s) (the result may still be extended or truncated as needed). These may also force the operands to obey their self-determined expression bit length.

- The concatenation operator is one such example of a self-determined expression with a bit-length that is well-defined as the sum of length of its operands, and in turn its operands are forced to use their self-determined expression length.

single-operand

{a} for which the result is the length of a

or multiple operands

e.g. {2'b00,b,a} for which the result length is 2+length(b)+length(a)

- The use of a single operand {} can force a self-determination for expressions like addition:

- In this next example, the self determined length of the addition is 8-bits which results in 0 for the summation. The 8-bit result from the concatenation operator is always unsigned and thus is zero extended.

```
ur8a = 128;
```

```
ur8b = 128;
```

```
ur16y= {ur8a+ur8b};
```

Rules for expression bit lengths from IEEE Standards Doc.

5.4.1 Rules for expression bit lengths

The rules governing the expression bit lengths have been formulated so that most practical situations have a natural solution.

The number of bits of an expression (known as the size of the expression) shall be determined by the operands involved in the expression and the context in which the expression is given.

A self-determined expression is one where the bit length of the expression is solely determined by the expression itself—for example, an expression representing a delay value.

A context-determined expression is one where the bit length of the expression is determined by the bit length of the expression and by the fact that it is part of another expression. For example, the bit size of the right-hand expression of an assignment depends on itself and the size of the left-hand side.

Table 5-22 shows how the form of an expression shall determine the bit lengths of the results of the expression. In Table 5-22, i , j , and k represent expressions of an operand, and $L(i)$ represents the bit length of the operand represented by i .

Multiplication may be performed without losing any overflow bits by assigning the result to something wide enough to hold it.

Reference from IEEE Standards Doc.

Table 5-22—Bit lengths resulting from self-determined expressions

Expression	Bit length	Comments
Unsize constant number ^a	Same as integer	
Sized constant number	As given	
i op j, where op is: + - * / % & ^ ^~ ~^	max(L(i),L(j))	
op i, where op is: + - ~	L(i)	
i op j, where op is: === != == != > >= < <=	1 bit	Operands are sized to max(L(i),L(j))
i op j, where op is: &&	1 bit	All operands are self-determined
op i, where op is: & ~& ~ ^ ~^ ~^!	1 bit	All operands are self-determined
i op j, where op is: >> << ** >>> <<<	L(i)	j is self-determined
i ? j : k	max(L(j),L(k))	i is self-determined
{i,...j}	L(i)+..+L(j)	All operands are self-determined
{i{j,...k}}	i * (L(j)+..+L(k))	All operands are self-determined

^aIf an unsize constant is part of an expression that is longer than 32 bits and if the most significant bit is unknown (X or x) or three-state (Z or z), the most significant bit is extended up to the size of the expression. Otherwise, signed constants are sign-extended and unsize constants are zero-extended.

Getting the Spec

At UMBC, goto <http://ieeexplore.ieee.org/>

Search for
IEEE Std 1364-2005

You'll find

IEEE Standard for Verilog Hardware Description
Language IEEE Std 1364-2005 (Revision of IEEE Std 1364-
2001)

Along with the older standard 1995 standard and the
SystemVerilog standard

Gotchas

```
. . .  
reg [3:0] remainingStock = 10; //unsigned  
  
always @ (posedge clk, negedge rst_)  
    if (rst_==0)  
        remainingStock<=10;  
    else if (remainingStock >= 0) //always TRUE!!!  
        remainingStock <= remainingStock-1;
```

Gotchas

```
. . .
input wire [2:0] remove;
signed reg [3:0] remainingStock = 10; //signed

always @ (posedige clk, negedge rst_)
    if (rst_==0)
        remainingStock<=10;
    else if ((remainingStock-remove) >= 0)
//always TRUE!!!
        remainingStock <= remainingStock-remove;
```

Scaling

Multiplication by positive integers may be performed with the shift operator if by a power of two:

Mult by 2^k is left shift by k

$a \ll k$ Typically synthesizers require k to be a constant;

k bits on left are discarded

Overflow Error: no overflow if all truncated bits are the same and the new msb matches the truncated bits

Logical Shifting is usually inexpensive – just “rewiring”.

Multiplication by numbers with a few non-zero bits can be inexpensive as well:

$$u * 24 = u \ll 4 + u \ll 3$$

This is an important tool for computer engineering to know.

Divide by power of two

Divide by 2^k is nearly arithmetic right shift

Need to Discard k bits on the right and replicate sign bit k times on the left. Must use the arithmetic shift;

$a \ggg k$; same as $\{k\{a[\text{msbindex}]\}, a[\text{msbindex}:k]\}$

However, integer division always drops the fractional bits.

This is also known as round towards zero

For a positive number this is $\text{floor}(x/n)$ $5/2=2$

For a negative number this is $\text{ciel}(x/n)$ $-5/2=-2$

If number is positive just use $a \ggg k$;

For a negative number, we want

$\text{ciel}(x/n)$ which may be computed as $\text{floor}((x+2^k-1)/(2^k))$

In verilog: $(a + (1 \ll k)) \ggg k$

Multiplication

In general to hold the result you need $M+N$ bits where M and N are the length of the operands

```
wire [N-1:0] a;  
wire [M-1:0] b;  
wire [M+N-1:0] y;
```

$y=a*b$; The multiplication in the context with the assignment will cause extension of operands to the size of the result